# Coupling low-order elements to high-order elements in ANSYS

# Aaron C Acton

20 June 2008

#### Abstract

This article presents several methods for obtaining coupling at the interface between low- and high-order elements in ANSYS. This situation can be encountered when modelling the fluid-structure interaction between a solid domain, modelled with quadratic elements, and an acoustic fluid domain, modelled with linear elements. The two methods determined to be most suitable for achieving this coupling are by using either constraint equations or contact elements, both of which are likely more-robust methods than by simply attaching coincident nodes on matching meshes. Examples are provided for each case described herein, and the results from each are compared to the ANSYS 11.0 verification problem VM177.

Keywords: Acoustics, Coupling, Constraint Equations, Contact Elements, Fluid-Structure Interaction

Tested in: ANSYS 11.0

#### 1 Introduction

When modelling with h-elements in ANSYS, users have the option of using low-order elements with linear shape functions or high-order elements with quadratic shape functions. The choice for using one or the other generally depends on geometry, material properties, and loading conditions; however, the options may be limited by element availability. In version 11.0 of ANSYS, for example, the 22x-series of coupled-field elements are available only in high-order form, and the acoustic or contained-fluid elements are available only in low-order form. This can cause problems when coupling between the low- and high-order elements is required, such as in the case of fluid-structure interaction (FSI). This article presents several different methods for achieving the coupling at the interface.

#### 2 Acoustics and Fluid-Structure Interaction

For FLUID29 and FLUID30 with a single degree of freedom (PRES) per node, a linear variation of acoustic pressure is assumed within the element. This is comparable to the use of quadratic shape functions in a displacement-formulated element [1]. For acoustic fluid-structure interaction, these acoustic elements offer a formulation with both displacement and pressure de-

grees of freedom, making the coupling at the interface relatively easy. As was previously mentioned, however, acoustic elements with quadratic shape functions are not available in ANSYS 11.0. For simulations in which the solid domain is modelled using high-order elements, the user is presented with the problem of coupling low-order elements to high-order elements.





Figure 1: Finite element model for fluid-structure interaction from VM177.

# 3 Finite Element Test Model

The ANSYS 11.0 Verification Manual test case VM177 [2] simulates a steel ring submerged in a com-

pressible fluid (water), represented by the finite-element model seen in Figure 1. While acoustic analyses typically simulate the outgoing effects of a pressure wave in a domain that extends to infinity [3], this analysis was modelled with a fluid domain that extends a finite radius from the solid. It was shown in the paper referenced by VM177 that a free vibration analysis of this geometry should result in an error of less than 1% compared to the frequency for an unbounded fluid.

The goal of this analysis was to find the lowest natural undamped frequency for the fluid-structure system bending in the plane of curvature (XY) and to validate the result in comparison to a reference value of 35.62 Hz, known to be analytically correct [2]. Since the mode of interest, specified in the problem description, was an extensional mode that was symmetric about the XZ and YZ planes, the analysis used a quarter-symmetry model to reduce the complexity and solution time. The displaced-structure plot for this geometry is shown in Figure 2, which also shows a resulting frequency of 36.545 Hz. This model will be modified and used as an example throughout this article.



**Figure 2:** Displaced-structure results for VM177 showing frequency of 36.545 Hz.

The examples presented here also verified that the circular ring alone (i.e., uncoupled from the fluid) yielded the correct natural frequency. ANSYS VM67 [2] was adapted to represent the geometry of the ring in VM177 and was used to calculate a numerical solution for the frequency (input file available in Appendix A.1). The model consisted of one axisymmetric-harmonic, 4-node, structural solid element and was analyzed to determine the second harmonic frequency in the plane of curvature (PLANE25 with MODE,2,1), which was the same extensional mode of interest in the coupled analyses.

An analytical solution was additionally calculated using an updated edition of the text referenced in VM67. Weaver [4] provides an equation for the frequency of the *i*th mode of vibration of a circular ring.

$$f_i = \frac{1}{2\pi} \sqrt{\frac{EI_z i^2 (1 - i^2)^2}{\rho A r^4 (1 + i^2)}}$$
 (1)

Substituting the geometric and material properties for the model in VM177 into Equation 1 gives a frequency of 61.986 Hz, while ANSYS gives a value of 61.971 Hz, a difference of 0.024%. Recalling that the coupled system yielded a frequency of 36.545 Hz, it becomes apparent that the fluid has a relatively-large effect on the natural frequency of ring when it is submerged in the fluid.

Although the focus of this article is coupling loworder elements to high-order elements, the case where both domains use low-order elements will be investigated first.

# 4 Low-Order, Similar Mesh

It is trivial to obtain a mesh in which the solid and fluid elements share nodes if the domains are first modelled using solid geometry. The volumes can simply be "glued" together using the VGLUE command, which redefines the input volumes so that they share areas along the common boundaries [5], and subsequent meshing of these volumes will automatically generate elements which share nodes at the boundary. In fact, if the volume keypoints are coincident, this same operation can be performed with the NUMMRG, KP command, an example of which can be seen in Appendix A.2. It may be necessary, however, to couple the nodes together after the model is meshed.




Figure 3: Low-order, similar mesh, coupling and FSI symbols shown.

When both domains use low-order elements and match up with similar meshes, the coupling can also be achieved simply by coupling coincident nodes.

While this is most efficiently accomplished by using the NUMMRG, NODE command, the same effect can be obtained by using the CPINTF command (see Appendix A.3 for input file). The latter method will be used here purely because the coupled nodes can be labelled in ANSYS using graphical symbols, as shown in Figure 3, whereas the merging-nodes method would be more difficult to present.

Coupling degrees of freedom into a set causes the results calculated for one member of the set to be the same for all members of the set. Instead of manually creating each coupled set, the CPINTF command can be used to "button" together elements interfacing at a seam, where the seam consists of a series of node pairs [5]. The coupled system in this example resulted in a frequency of 35.682 Hz, while the ring alone yielded 62.404 Hz.

When the solid domain is modelled using high-order elements and the meshes do not match, these simple methods of node merging or coupling become invalid since there are no coincident nodes to couple and since the element order is different. Fortunately, ANSYS provides at least two features which can be used to accomplish the coupling, namely constraint equations and contact elements.




Figure 4: High-order, dissimilar mesh, constraint-equation method, coupling and FSI symbols shown.

# 5 High-Order, Dissimilar Mesh, Constraints

Linear constraint equations may be used to relate the degrees of freedom of selected nodes in a more general manner than the nodal coupling described in Section 4. Instead of manually creating each constraint equation, the CEINTF command can be used to "tie" together two regions with dissimilar mesh patterns by generating constraint equations that connect the selected nodes of one region to the selected elements of the other region [5]. A dissimilar mesh pattern was

purposely created for this example, shown in Figure 4, and the domains were coupled together using constraint equations. The coupled system in this example resulted in a frequency of 35.683 Hz, while the ring alone yielded 62.294 Hz. This input file used can be found in Appendix A.4.

# 6 High-Order, Dissimilar Mesh, Contact

Contact elements may also be used to relate the motion of two flexible surfaces; a "target surface" of TARGE170 elements and a "contact surface" of CONTA174 elements form a "contact pair," while the ESURF command can be used to generate the contact and target elements along the boundaries of an existing mesh. A discussion of all of the contact settings will not be included here as the topic can be relatively complicated, but for the purposes of this example, MPC-type contact will be used as it works similarly to the CEINTF command in for small deformation analysis (NLGEOM, OFF) [6]. A dissimilar mesh pattern was again purposely created for this example and can be seen in Figure 5. The frequency of the coupled system and of the ring alone in this example were the same as those in Section 5, 35.683 and 62.294 Hz, respectively. The input file for this example can be found in Appendix A.5.





Figure 5: High-order, dissimilar mesh, contact-element method, coupling and FSI symbols shown.

Penalty-based contact methods (pure penalty and augmented Lagrange) are also applicable to this type of analysis, but these formulations generally calculate a normal stiffness value (FKN) based on the elastic modulus (MP,EX) of the underlying elements. Since acoustic elements have no elastic modulus, the user is presented with the problem of manually setting an appropriate normal stiffness. This can simply be avoided by using the MPC-type contact, as it does not require a normal stiffness value.

# 7 Alternative Approaches


In addition to the methods described above, it is possible to approach this problem in at least three other ways.



**Figure 6:** High-order, similar mesh, coupling and FSI symbols shown.

#### 7.1 High-Order, Similar Mesh, Coupled Sets

For the case where the meshes from both domains match, corner nodes on the solid elements are coincident with those on the fluid elements. The solid-element midside nodes, however, have no pairing on the fluid domain. In this situation, it may seem as though the corner nodes could be coupled using either the NUMMRG, NODE or the CPINTF command, while leaving the midside nodes uncoupled, as shown in Figure 6. In fact, when an example was tested for this article (see Appendix A.6 and A.7), the resulting frequency was 35.728 Hz, a close match to the reference.



**Figure 7:** Equivalent nodal distribution of a unit uniform surface load (source: Modeling and Meshing Guide [7]).

While this is perhaps the simplest approach, it is not necessarily a good practice in general. The corner nodes of quadratic elements (e.g., 20-node brick SOLID185) are associated with negative stiffness values, such that

a unit uniform surface load is applied with the non-intuitive nodal distribution shown in Figure 7. This results in an "effective stiffness" at the surface that is very non-uniform [6], and consequently, coupling high-order elements directly to low-order elements in this manner can lead to convergence difficulties in nonlinear analyses.




Figure 8: High-order, similar mesh, coarse solid, coupling and FSI symbols shown.




Figure 9: Resulting displacement of the uncoupled node on the acoustic elements (solid elements hidden).

# 7.2 High-Order, Coarse Solid Mesh, Coupled Sets

The solid and fluid meshes may match up when the solid-element edge lengths are exactly double that of the fluid elements (Appendix A.8 contains an example). In this case, the solid-element midside nodes coincide with some of the fluid-element corner nodes, such that four fluid elements could attach to each solid element with the exception of one node. If coupling is attempted despite this mismatch, as shown in Figure 8, the resulting interaction will not be accurately represented, which can be seen by the increased natural frequency of 67.198 Hz, as well as the errant displacement of the uncoupled node shown in Figure 9.

### 7.3 High-Order, Dropped Nodes, Coupled Sets

When the meshes match up and the solid is modelled using quadratic elements, it may be of interest to obtain a linear shape function at the interface with the fluid. Solid-element midside nodes can be added or removed by using the EMID, ADD or EMID, REMOVE commands, respectively. Dropping the midside nodes on the solid interface will reduce the order of the shape function for that face, as shown in Figure 10; however, the high-order elements with dropped midside nodes behave in a "stiff" manner, resulting in a higher natural frequency of 86.463 Hz in this example (input file available in Appendix A.9).




Figure 10: High-order, similar mesh, dropped midside nodes, coupling and FSI symbols shown.

This behaviour occurs because the quadratic elements do not have an enhanced-strain formulation like the linear elements, making them susceptible to "shear locking." This numerical phenomenon is known to increase the flexural stiffness of solid structures, hence the reference to being "stiff" and the much higher resulting frequency. Figure 11 shows two element-contour plots of  $r\theta$ -shear stress (PLESOL,S,XY with RSYS,1); the MPC-contact example presented in Section 6 results in a smooth stress pattern, while this dropped-nodes example results in a fluctuating stress pattern, which is often indicative of shear locking. It should be noted, however, that pure "stretching" or "extending" modes will not be affected by shear locking.

Alternatively, a mesh in which quadratic-element midside nodes are removed can be obtained without explicitly using the EMID, REMOVE command if the domain volumes share a common boundary. When one volume (or area) is meshed with linear elements, then an adjacent volume (or area) is meshed with quadratic elements, ANSYS will automatically remove midside nodes along the common sides of linear and quadratic

elements [7]. The output gives a note that

SOLID186 element 481 (and possibly others) does not have all of its midside nodes supplied. In some situations, this may reduce solution accuracy.

An input file for this example can be found in Appendix A.10.

#### 8 Results and Discussion

Table 1 summarizes the results from the above examples. For each example, the natural frequency from the coupled system is compared to the reference value in VM177, and that from the uncoupled structure is compared to a benchmark value obtained by adapting VM67 to this geometry. All of the input files are available in the appendices. It can be seen from the results that the constraint-equation method and the contact-element method both resulted in accurate results.

For the other three examples, it was determined that none of the methods were suitable for achieving coupling, and consequently, the error was not calculated for these cases as a visual indication that while the frequency may have matched up in these examples, it is not necessarily a suggested method of coupling in general.


#### 9 Conclusions


Coupling can be achieved between low-order elements and high-order elements in ANSYS in several ways, but it was found that the constraint-equation method and the contact-element method were most effective. There are also several alternative approaches to attempting this coupling that may not obtain the desired result. These methods involve either excluding the solid-element midside nodes from the coupled set, excluding one acoustic node from the coupled set, or dropping the solid-element midside nodes.

#### References

- [1] D. B. Woyak, "Acoustics and fluid-structure interaction," 1992.
- [2] ANSYS, Inc., "Verification Manual: ANSYS Release 11.0," 2007.
- [3] ANSYS, Inc., "Fluids Analysis Guide: ANSYS Release 11.0," 2007.

- [4] W. Weaver, Jr., S. P. Timoshenko, and D. H. Young, Vibration Problems in Engineering. John Wiley & Sons, 5th ed., 1990.
- [5] ANSYS, Inc., "Commands Reference: ANSYS Release 11.0," 2007.
- [6] ANSYS, Inc., "Contact Technology Guide: ANSYS Release 11.0," 2007.
- [7] ANSYS, Inc., "Modeling and Meshing Guide: ANSYS Release 11.0," 2007.





- (a) Smooth pattern of element  $r\theta\text{-shear}$  stress in solid domain.
- (b) Fluctuating pattern of element  $r\theta$ -shear stress in solid domain.

**Figure 11:** Difference in results between example using MPC contact case and example using dropped midside nodes.

**Table 1:** Comparison of results for first natural frequency for different coupling methods and for ring only.

	Coupled System		Ring Only	
Model	Freq [Hz]	Err [%]	Freq [Hz]	Err [%]
Reference	35.62	_	61.986	_
VM177	36.545	2.6	62.278	0.47
Low, Similar, CPINTF	35.682	0.17	62.404	0.67
High, Dissimilar, CEINTF	35.683	0.18	62.294	0.50
High, Dissimilar, MPC	35.683	0.18	62.294	0.50
High, Similar, CPINTF	35.728	_	62.374	_
High, Coarse, CPINTF	67.198	_	67.317	_
High, Dropped, CPINTF	86.463	_	140.61	_

# A ANSYS Input Files

All input files used for the examples presented in this article are included in the following sections. Header information and page numbering has been removed to simplify copy-and-paste operations.

# A.1 Calculation of natural frequency for bending mode of ring only

```
FINISH
/CLEAR
/PREP7
ET,1,PLANE25
MP, EX, 1, 30. E6
MP, NUXY, 1, 0.0
MP, DENS, 1, 7.4167E-4
N,1,10+0.25/2,0
N,2,10+0.25/2,1
N,3,10-0.25/2,1
N,4,10-0.25/2,0
E,1,2,3,4
D,1,UY
CP,1,UX,3,4
FINISH
/SOLU
ANTYPE, MODAL
MODOPT, LANB, 1
MODE, 2, 1
SOLVE
FINISH
/POST1
SET,1,1
PLDISP
```

# A.2 Low-order solid, Similar mesh

```
FINISH
/CLEAR
/TITLE,LOW-ORDER SOLID, SIMILAR MESH, NUMMRG,KP METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
                                ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,1,FLUID30
ET,2,SOLID185,,3
                                ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                                ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP, DENS, 1, 9.6333E-5
MP,SONC,1,57480.0
                                 ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
MP, NUXY, 2, 0.3
CSYS,1
! DEFINE ACOUSTIC DOMAIN
{\tt CYLIND,my\_a+my\_t/2,my\_b,0,1,0,90}
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
```

```
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,O.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, ALL 4 NODES ON EACH ELEMENT ARE CONNECTED
ALLSEL, ALL
NUMMRG, KP
VMESH, ALL
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE,3
REAL,1
MAT,1
EMODIF, ALL
ALLSEL, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
                                ! DEFINE FLUID-STRUCTURE INTERFACE
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
SF, ALL, FSI
ESEL.ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D, ALL, UX
NSEL,S,LOC,Y,O.O
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D, ALL, PRES, 0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                               ! MODAL ANALYSIS
MODOPT, UNSYM, 4,-100
                                ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
!ESEL,S,TYPE,,2
SOLVE
FINISH
/POST1
```

SET,1,2

# A.3 Low-order solid, Similar mesh, CPINTF method

```
FINISH
/CLEAR
/TITLE,LOW-ORDER SOLID, SIMILAR MESH, CPINTF METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                               ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID185,,3
                                ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                                ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP,DENS,1,9.6333E-5
MP,SONC,1,57480.0
                                ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
CSYS,1
! DEFINE ACOUSTIC DOMAIN
CYLIND, my_a+my_t/2, my_b, 0, 1, 0, 90
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
ALLSEL, ALL
VMESH, ALL
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
{\tt NSEL,S,LOC,X,my\_a+my\_t/2}
ESLN,U,O,ALL
TYPE,3
REAL, 1
MAT,1
EMODIF, ALL
ALLSEL, ALL
```

! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN

! IN THIS CASE, ALL 4 NODES ON EACH ELEMENT ARE CONNECTED

```
!NUMMRG, NODE ! either NUMMRG or CPINTF
CPINTF, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
ESEL,S,TYPE,,1
                               ! DEFINE FLUID-STRUCTURE INTERFACE
NSEL,S,LOC,X,my_a+my_t/2
SF, ALL, FSI
ESEL, ALL
NSEL.ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D,ALL,UX
NSEL,S,LOC,Y,0.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D, ALL, UY
NSEL.ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D, ALL, PRES, 0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                               ! MODAL ANALYSIS
MODOPT, UNSYM, 4, -100
                               ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
!ESEL,S,TYPE,,2
SOLVE
FINISH
/POST1
SET,1,2
PLDISP
A.4 High-order solid, Dissimilar mesh, CEINTF method
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, DISSIMILAR MESH, CEINTF METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                              ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID186,,1
                               ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                               ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP,DENS,1,9.6333E-5
MP,SONC,1,57480.0
                                ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
CSYS,1
! DEFINE ACOUSTIC DOMAIN
CYLIND, my_a+my_t/2, my_b, 0, 1, 0, 90
```

VSEL,S,,,1 VATT,1,1,1

```
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang+1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
ALLSEL, ALL
VMESH, ALL
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE,3
REAL, 1
MAT,1
EMODIF, ALL
ALLSEL, ALL
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, THE ELEMENTS FROM THE STUCTURAL DOMAIN
! ARE COUPLED TO THE NODOES FROM THE ACOUSTIC DOMAIN
NSEL,S,LOC,X,my_a+my_t/2
ESLN,S,0
ESEL,R,TYPE,,2
CM, STRUCT, ELEM
ESEL,S,TYPE,,1
NSLE,S,ALL
NSEL,R,LOC,X,my_a+my_t/2
CM, ACOUST, NODE
CMSEL,S,STRUCT
CMSEL,S,ACOUST
CEINTF
ALLSEL, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
ESEL,S,TYPE,,1
                                ! DEFINE FLUID-STRUCTURE INTERFACE
NSEL,S,LOC,X,my_a+my_t/2
SF, ALL, FSI
ESEL, ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D,ALL,UX
NSEL,S,LOC,Y,0.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
```

NSEL,S,LOC,X,my\_b

```
D, ALL, PRES, 0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                               ! MODAL ANALYSIS
                                ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
MODOPT, UNSYM, 4,-100
!ESEL,S,TYPE,,2
SOLVE
FINISH
/POST1
SET,1,2
PLDISP
A.5 High-order solid, Dissimilar mesh, Contact method
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, DISSIMILAR MESH, CONTACT METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                               ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID186,,1
                               ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                               ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP, DENS, 1, 9.6333E-5
                               ! SPEED OF SOUND IN WATER
MP,SONC,1,57480.0
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
CSYS.1
! DEFINE ACOUSTIC DOMAIN
\texttt{CYLIND,my\_a+my\_t/2,my\_b,0,1,0,90}
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang+1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
```

ALLSEL, ALL VMESH, ALL

```
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE,3
REAL,1
MAT,1
EMODIF, ALL
ALLSEL, ALL
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, BONDED CONTACT IS USED
ET,4,170
ET,5,174
KEYOPT,5,4,2
KEYOPT,5,2,2
KEYOPT,5,12,5
MAT,4
REAL,4
NSEL,S,LOC,X,my_a+my_t/2
ESLN,S,O
ESEL,R,TYPE,,2
TYPE,4
ESURF
NSEL,S,LOC,X,my_a+my_t/2
ESLN,S,0
ESEL,R,TYPE,,1
TYPE,5
ESURF
ALLSEL, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
ESEL,S,TYPE,,1
                               ! DEFINE FLUID-STRUCTURE INTERFACE
NSEL,S,LOC,X,my_a+my_t/2
SF, ALL, FSI
ESEL, ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D,ALL,UX
NSEL,S,LOC,Y,0.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D,ALL,PRES,0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                                ! MODAL ANALYSIS
MODOPT, UNSYM, 4, -100
                                ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
!ESEL,S,TYPE,,2
SOLVE
/show,png
eplot
/show,close
/show,term
FINISH
```

/POST1 SET,1,2

### A.6 High-order solid, Similar mesh, NUMMRG,KP method

```
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, SIMILAR MESH, NUMMRG, KP METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                               ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID186,,1
                               ! SOLID ELEMENTS TO MODEL STEEL RING
                                ! NON-INTERFACING FLUID ELEMENTS
ET,3,FLUID30,,1
R,1,1
MP, DENS, 1, 9.6333E-5
MP,SONC,1,57480.0
                                ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
MP, NUXY, 2, 0.3
CSYS,1
! DEFINE ACOUSTIC DOMAIN
CYLIND, my_a+my_t/2, my_b, 0, 1, 0, 90
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, THE MIDSIDE NODES ARE NOT DROPPED AUTOMATICALLY
ALLSEL, ALL
NUMMRG, KP
VMESH,2
                                ! VOLUME WITH QUADRATIC ELEMENTS IS MESHED FIRST
VMESH,1
                                ! FOLLOWED BY VOLUME WITH LINEAR ELEMENTS
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE.3
REAL, 1
MAT,1
```

```
EMODIF, ALL
ALLSEL, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
                                ! DEFINE FLUID-STRUCTURE INTERFACE
ESEL,S,TYPE,,1
{\tt NSEL,S,LOC,X,my\_a+my\_t/2}
SF, ALL, FSI
ESEL, ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D,ALL,UX
NSEL,S,LOC,Y,0.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D, ALL, PRES, 0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                               ! MODAL ANALYSIS
MODOPT, UNSYM, 4,-100
                                ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
!ESEL,S,TYPE,,2
SOLVE
FINISH
/POST1
SET,1,2
PLDISP
A.7 High-order solid, Similar mesh, CPINTF method
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, SIMILAR MESH, CPINTF METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                               ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID186,,1
                               ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                                ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP,DENS,1,9.6333E-5
MP,SONC,1,57480.0
                                ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
CSYS,1
! DEFINE ACOUSTIC DOMAIN
CYLIND, my_a+my_t/2, my_b, 0, 1, 0, 90
```

VSEL,S,,,1 VATT,1,1,1

```
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
ALLSEL, ALL
VMESH, ALL
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE,3
REAL,1
MAT,1
EMODIF, ALL
ALLSEL, ALL
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, ONLY THE 4 CORNER NODES ON EACH ELEMENT ARE CONNECTED
!NUMMRG,NODE ! either NUMMRG or CPINTF
CPINTF, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
                                ! DEFINE FLUID-STRUCTURE INTERFACE
ESEL,S,TYPE,,1
{\tt NSEL,S,LOC,X,my\_a+my\_t/2}
SF, ALL, FSI
ESEL, ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D, ALL, UX
NSEL,S,LOC,Y,0.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D, ALL, PRES, 0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                                ! MODAL ANALYSIS
                                ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
MODOPT, UNSYM, 4,-100
!ESEL,S,TYPE,,2
SOLVE
```

FINISH

ALLSEL, ALL

# A.8 High-order solid, Similar mesh, Coarse solid, CPINTF method

```
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, SIMILAR MESH, COARSE SOLID, CPINTF METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                                ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID186,,1
                                ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                               ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP,DENS,1,9.6333E-5
MP,SONC,1,57480.0
                                ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
CSYS,1
! DEFINE ACOUSTIC DOMAIN
\texttt{CYLIND,my\_a+my\_t/2,my\_b,0,1,0,90}
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang/2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick/2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick
ALLSEL, ALL
VMESH, ALL
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE,3
REAL,1
MAT,1
EMODIF, ALL
```

```
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, 4 ACOUSTIC ELEMENTS ARE CONNECTED TO EACH STRUCTURAL ELEMENT
!NUMMRG,NODE ! either NUMMRG or CPINTF
CPINTF, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
ESEL,S,TYPE,,1
                              ! DEFINE FLUID-STRUCTURE INTERFACE
NSEL,S,LOC,X,my_a+my_t/2
SF, ALL, FSI
ESEL, ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D,ALL,UX
NSEL,S,LOC,Y,0.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D,ALL,PRES,0.0
NSEL, ALL
FINISH
/SOLU
ANTYPE, MODAL
                               ! MODAL ANALYSIS
MODOPT, UNSYM, 4,-100
                               ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
!ESEL,S,TYPE,,2
SOLVE
FINISH
/POST1
SET,1,2
PLDISP
     High-order solid, Similar mesh, Dropped midside nodes, CPINTF method
A.9
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, SIMILAR MESH, DROPPED MIDSIDE NODES, CPINTF METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                               ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
                               ! SOLID ELEMENTS TO MODEL STEEL RING
ET,2,SOLID186,,1
ET,3,FLUID30,,1
                               ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP,DENS,1,9.6333E-5
                               ! SPEED OF SOUND IN WATER
MP,SONC,1,57480.0
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
CSYS,1
! DEFINE ACOUSTIC DOMAIN
```

CYLIND, my\_a+my\_t/2, my\_b, 0, 1, 0, 90

```
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
CYLIND, my_a-my_t/2, my_a+my_t/2,0,1,0,90
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE,ALL,,,ndiv_thick*2
ALLSEL, ALL
VMESH, ALL
! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
ESLN,U,O,ALL
TYPE,3
REAL,1
MAT,1
EMODIF, ALL
ALLSEL, ALL
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, 4 NODES ON EACH SOLID ELEMENT ARE DROPPED MANUALLY
CPINTF, ALL
ESEL,S,TYPE,,2
NSLE,S,ALL
NSEL,R,LOC,X,my_a+my_t/2
MODMSH, DETACH
EMID, REMOVE, BOTH
ALLSEL, ALL
! COUPLE STRUCTURAL MOTION & FLUID PRESS.
                                ! DEFINE FLUID-STRUCTURE INTERFACE
ESEL,S,TYPE,,1
NSEL,S,LOC,X,my_a+my_t/2
SF, ALL, FSI
ESEL, ALL
NSEL, ALL
! SET SYMMETRIC BOUNDARY CONDITION
NSEL,S,LOC,Y,90.0
NSEL,R,LOC,X,my_a-my_t/2,my_a+my_t/2
D, ALL, UX
NSEL,S,LOC,Y,0.0
{\tt NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2}
D, ALL, UY
NSEL, ALL
! SET PRESSURE AT OUTER RADIUS TO ZERO
NSEL,S,LOC,X,my_b
D,ALL,PRES,0.0
NSEL, ALL
```

FINISH

```
ANTYPE, MODAL
                                ! MODAL ANALYSIS
MODOPT, UNSYM, 4, -100
                                ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION
!ESEL,S,TYPE,,2
SOLVE
FINISH
/POST1
SET,1,2
PLDISP
A.10 High-order solid, Similar mesh, Dropped midside nodes
FINISH
/CLEAR
/TITLE, HIGH-ORDER SOLID, SIMILAR MESH, DROPPED MIDSIDE NODES, NUMMRG, KP METHOD
my_a = 10
my_b = 30
my_t = 0.25
ndiv_tang = 16
ndiv_thick = 2
/PREP7
ET,1,FLUID30
                               ! FLUID ELEMENTS INTERFACING WITH STRUCTURE
ET,2,SOLID186,,1
                               ! SOLID ELEMENTS TO MODEL STEEL RING
ET,3,FLUID30,,1
                                ! NON-INTERFACING FLUID ELEMENTS
R,1,1
MP, DENS, 1, 9.6333E-5
MP,SONC,1,57480.0
                                ! SPEED OF SOUND IN WATER
MP,EX,2,30.E6
MP, NUXY, 2, 0.3
MP, DENS, 2, 7.4167E-4
MP, NUXY, 2, 0.3
CSYS,1
! DEFINE ACOUSTIC DOMAIN
CYLIND, my_a+my_t/2, my_b, 0, 1, 0, 90
VSEL,S,,,1
VATT,1,1,1
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
! DEFINE SOLID DOMAIN
\texttt{CYLIND,my\_a-my\_t/2,my\_a+my\_t/2,0,1,0,90}
VSEL,S,,,2
VATT,2,2,2
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Y,45
LESIZE, ALL,,,ndiv_tang
ALLSEL, BELOW, VOLU
LSEL,R,LOC,Z,0.5
LESIZE, ALL,,,ndiv_thick
ALLSEL, BELOW, VOLU
LSEL,R,LOC,X,my_a
LESIZE, ALL,,,ndiv_thick*2
! ATTACH STRUCTURAL DOMAIN TO FLUID DOMAIN
! IN THIS CASE, 4 NODES ON EACH SOLID ELEMENT ARE DROPPED AUTOMATICALLY
ALLSEL, ALL
NUMMRG, KP
                                ! VOLUME WITH LINEAR ELEMENTS IS MESHED FIRST
VMESH,1
```

/SOLU

```
VMESH,2
```

#### ! FOLLOWED BY VOLUME WITH QUADRATIC ELEMENTS

### ! MODIFY ACOUSTIC DOMAIN FOR NON-INTERFACING ELEMENTS

ESEL,S,TYPE,,1
NSEL,S,LOC,X,my\_a+my\_t/2
ESLN,U,O,ALL
TYPE,3
REAL,1
MAT,1
EMODIF,ALL
ALLSEL,ALL

#### ! COUPLE STRUCTURAL MOTION & FLUID PRESS.

ESEL,S,TYPE,,1 ! DEFINE FLUID-STRUCTURE INTERFACE
NSEL,S,LOC,X,my\_a+my\_t/2
SF,ALL,FSI
ESEL,ALL
NSEL,ALL

#### ! SET SYMMETRIC BOUNDARY CONDITION

NSEL,S,LOC,Y,90.0 NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2 D,ALL,UX NSEL,S,LOC,Y,0.0 NSEL,R,LOC,X,my\_a-my\_t/2,my\_a+my\_t/2 D,ALL,UY NSEL,ALL

# ! SET PRESSURE AT OUTER RADIUS TO ZERO

NSEL,S,LOC,X,my\_b D,ALL,PRES,0.0 NSEL,ALL

FINISH

/SOLU

ANTYPE, MODAL ! MODAL ANALYSIS

MODOPT, UNSYM, 4,-100 ! SELECT UNSYMMETRIC MATRIX MODE EXTRACTION

!ESEL,S,TYPE,,2

SOLVE FINISH

/POST1 SET,1,2 PLDISP