Collaborative Solutions Inc

An ANSYS Support Distributor

Date February 27, 2000 Memo Number STI35:000227A
Subject ANSYS Tips & Tricks: Structural Surface Effect Elements
Keywords Surface Effect: Structural

1. Introduction:

Surface effect elements are extremely useful tools in both structural and thermal analyses. These elements provide the user with more flexibility in applying loads to their model. This memo is part 1 in a series which hopes to provide users with more information on the structural and thermal surface effect elements, SURF151-154 in ANSYS 5.5 and 5.6 (similar to SURF19 and SURF22 in ANSYS 5.4).

For the present discussion, the basics of structural surface effect elements will be reviewed.

2. Background Discussion:

Surface effect elements have no physical properties. These elements are mainly used for loading purposes only. They are overlaid like a "skin" on structural (and thermal) element faces.

Structural surface effect elements, SURF153/154, allow the user to define pressures in various manners:

- Application of a tangent pressure such as traction or an overall moment
- Pressure on a projected area such as a bolt load
- Pressure with an arbitrary direction

The "regular" pressure loading applied via the SFx commands only act normal to the face, so the above uses of the structural surface effect elements provide much more flexibility with pressure loads.

3. Example of Generating Surface Effect Elements:

The SURF153/154 elements can be used with any lower- and higher-order structural element, with the exception of axisymmetric-harmonic elements (PLANE25, PLANE83, FLUID81). These surface effect elements are generated on top of existing element faces via ESURF or LMESH/AMESH commands (or direct generation).

For example, the below input file illustrates the two methods of generating surface effect elements:

```
! ESURF elements
/prep7
et,1,95
et,2,154
block,,10,,10,,10
vatt,1,1,1
esize,1
vmesh,all
asel,s,loc,y,10
nsla,s,1        ! select nodes for ESURF
esel,all        ! select elements for ESURF
type,2
esurf
```

```
! AMESH elements
/prep7
et,1,95
et,2,154
block,,10,,10,,10
vatt,1,1,1
esize,1
vmesh,all
asel,s,loc,y,10
type,2
amesh,all
```

As noted in the example, the left-hand side generates surface effect elements via the ESURF command, which directly generates them from selected nodes. ANSYS looks for selected nodes and elements on which to overlay surface effect elements. The selected nodes tell ANSYS which nodes to use to generate surface effect elements. The selected elements tell ANSYS which element faces to use (corresponding to the selected nodes) and how to orient the element z-axis (pointing away from solid element). Usually, selecting the nodes is most important, and the user can issue "ESEL,ALL" for the elements – ANSYS will not generate SURF153/154 on all the elements but only those faces whose nodes are selected.

The right-hand input shows another way of generating surface effect elements. Because the volume is meshed first, ANSYS knows how to orient the surface effect elements (element z-axis pointing away from volume). Unlike the ESURF command, AMESH allows the surface effect elements to retain associativity with the areas, so instead of EDELE, the user must issue ACLEAR to delete the surface effect elements.

Neither of the above methods is necessarily "better" than the other; it is simply a matter of preference for the user in generating surface effect elements.

4. Example of Traction Load:

Below is an input file demonstrating the use of surface effect elements to generate a traction load on a simple block. A lower-order SOLID185 is used, and the load is applied as an arbitrary vector:

```
/title,Block w/ Traction
/view,1,1,1,1
/auto
/plopt,info,on
/plopt,leg2,off
/plopt,minm,off
/triad,lbot

*set,HEIGHT,10
*set,WIDTH ,10
*set,DEPTH ,2
/prep7
et,1,185
r,1,
mp,ex  ,1,10e6
mp,nuxy,1,.3
mp,dens,1,.1/386.4

et,2,154
keyopt,2,4,1
keyopt,2,11,2
r,2,
mp,dens,2,0.0

block,,WIDTH,,HEIGHT,,DEPTH
vatt,1,1,1
esize,1
mshape,0
mshkey,1
vmesh,all

asel,s,loc,y,0
da,all,ux

asel,s,loc,x,0
da,all,uy

asel,s,loc,z,0
da,all,uz

asel,s,loc,y,HEIGHT
asel,a,loc,x,WIDTH
nsla,s,1
type,2
real,2
mat ,2
esurf,all

nsel,s,loc,x,WIDTH
esln,s,1
sfe,all,5,pres,,100,,1

nsel,s,loc,y,HEIGHT
esln,s,1
sfe,all,5,pres,,100,1,

allsel,all
/psf,pres,norm,2,1
/pbc,u,1
sbctran
eplot
/wait,5

finish

/solu
antype,static,new
eqslv,pcg,1e-8
solve
finish

/post1
set,last
/view,1,,,1
/title,von  Mises  Stress:  sqrt(3)  times
Applied Pressure of 100 psi
plnsol,s,eqv,2
/wait,5
/title,Shear  Stress:  Same  as  Applied
Pressure of 100 psi
plnsol,s,xy,2
finish
```

The figure on the right shows the block with boundary conditions and loads applied. Only the surface effect elements are shown for clarity (the SOLID185 elements are unselected).



The boundary conditions are shown in blue; only constraint in the UZ direction was applied to prevent rigid-body motion.

In this specific case, a pressure was applied to the surface effect elements on face 5, which represent an arbitrary load vector. The load magnitude is applied as the VAL1 argument. The arguments VAL2, VAL3, and VAL4 serve to provide a vector for the direction of the load. In this case, the right-side has a load vector of (0,1,0) and the top has a load vector of (1,0,0) as shown in the input file.

The results show a constant shear stress of 100 psi, as expected. The reaction force is also as expected, pressure*area.

5. Example of Moment Load:

An input file is presented which creates a simple rod and applies a moment on the top via surface effect elements. Higher-order SOLID95 elements are used, and the load is applied along an element coordinate system:

```
!----------------------------------------
! Input torque applied (moment)
! Input radius, height, element size...
!----------------------------------------
TORQUE   = 100
RADIUS   = 2
H_TIP    = 2
HEIGHT   = 20
ELEMSIZE = 1

PI       = acos(-1)
FORCE    = 100/RADIUS
PRESSURE = FORCE/(H_TIP*2*PI*RADIUS)

/graphics,power
/efacet,2
/view,1,1,1,1
/auto
/triad,rbot
/plopt,leg2,off
/plopt,info,on

!----------------------------------------
! Define higher-order SOLID95
! Define surface effect elements SURF154
!   which is used to apply torque
!   as a tangential pressure
!----------------------------------------
/prep7
et,1,95
et,2,154
r,1,
r,2,

!----------------------------------------
! Aluminum properties (or something)
!----------------------------------------
mp,ex,1,10e6
mp,nuxy,1,.3
mp,dens,1,.1/386.1
mp,dens,2,0

!----------------------------------------
! Simple cylinder
!----------------------------------------
*do,ICOUNT,1,4
  cylind,RADIUS,,HEIGHT-
H_TIP,HEIGHT,90*(ICOUNT-1),90*ICOUNT
  cylind,RADIUS,,HEIGHT-H_TIP,,90*(ICOUNT-
1),90*ICOUNT
*enddo

nummrg,kp

lsel,s,loc,x,0
lsel,r,loc,y,0
lsel,r,loc,z,0,HEIGHT-H_TIP
lesize,all,ELEMSIZE*2
mshape,0
mshkey,1
esize,ELEMSIZE
allsel,all
vmesh,all

csys,1
asel,s,loc,z,HEIGHT-H_TIP+0.0001,HEIGHT-
0.0001
asel,r,loc,x,RADIUS
local,11,1
csys,0
aatt,2,2,2,11
amesh,all
finish

/solu
antype,static,new
eqslv,pcg,1e-8

!----------------------------------------
! Apply tangential pressure
!----------------------------------------
esel,s,type,,2
sfe,all,2,pres,,PRESSURE

!----------------------------------------
! Constrain bottom of cylinder/rod
!----------------------------------------
asel,s,loc,z,0
nsla,s,1
d,all,all

allsel,all
/psf,pres,,2
/pbc,u,1
/title,Loads  and  boundary  conditions  on
model
eplot
*set,ENTER,0.0
*ask,NEXT,Press [Enter] to Continue,ENTER
/title,Simple torsional example
solve
finish

/post1
set,last
fsum
esel,u,type,,2
/edge,1,1
plnsol,s,eqv
```

Please note that in this input file, due to the size of the columns, some of the input lines are too long and are wrapped around.





The resulting model is shown on the right. As mentioned above, it is a simple rod with a moment applied at an end. The bottom is constrained fully.

This moment is applied with surface effect elements. The surface effect elements are shown with element coordinate system symbols on. The element coordinate systems of the surface effect elements were rotated to the Global Cylindrical coordinate system (element x-axis in global cylindrical y-direction).

The surface loads (SFx family of commands) make use of load keys (usually corresponding with element faces). For the SURF154 elements, a pressure load on face 5 corresponds to an arbitrary load as shown in the traction example above. For this case, we used a pressure load on face 2, which corresponds with a pressure aligned in the direction of the element x-axis. This allows us to apply a traction tangent to the surface, creating a moment.

After running the analysis, the FSUM command provides force and moment summation about a point defined with the SPOINT command:

```
***** SUMMATION OF TOTAL    FORCES AND MOMENTS IN
GLOBAL COORDINATES *****
  FX   =  0.1831868E-14
  FY   = -0.1054712E-14
  FZ   =  0.8770990E-14
  MX   =  0.2202682E-12
  MY   = -0.6217251E-13
  MZ   =   99.99507

 SUMMATION POINT=  0.0000       0.0000       0.0000
```

This shows that the moment of 100 lb-in was applied correctly on the model. The below plots show displacement results (contour and vector plots, respectively).





6. Example of Bolt Load:

The surface effect elements provide various options to mimic special types of loads such as bolt-type loading. For example, consider a simple block with a hole as shown in the figures below:



The model on the right is shown, meshed with SOLID45 elements. SURF154 elements are meshed in the cylindrical area defining the hole. A clearer view of the elements is shown on the left (in purple).

The surface effect elements allow various options in considering the tangential and normal components of pressure. Projected or full area of application can be changed as well.

For a bolt load, a user may want to apply the load on a projected area only, with the tangential components of load. Moreover, "negative" pressures are undesirable ("pulling" the surface). These options are controlled via the various keyoptions:

- KEYOPT(11) determines whether the projected or full area is considered. Moreover, the tangential component can be included (or excluded). For this model, KEYOPT(11)=0
- KEYOPT(12) controls whether or not "negative" pressures are considered. In this case, KEYOPT(12)=1

In this manner, the pressure loading can be controlled to mimic specific types of behavior. It is important to note that some keyoptions only correspond to given loads. In this case, we are using a pressure load key of 5 (arbitrary load vector), so keyoptions 11 and 12 are applicable.

Lastly, the pressure is applied to the surface effect elements only, with load key (face) 5. The direction of the load is shown with the red arrows. In /POST1, the magnitude of the load can be postprocessed via element tables (ETABLE) as shown above.



The resulting stress distribution is shown on the right. The bolt load provides a parabolic load distribution as desired. The stresses and deformation are as expected for this type of loading. Also, verification of the reaction forces reveal that it is equal to (pressure) * (projected area of bolt hole).

It may be instructive to note that this example is for illustrative purposes only. The author does not wish to present this as a "recommended" way of modeling bolt loads. There are other, more representative methods of applying similar loads. This scenario is presented as a means to illustrate some of the capabilities of the surface effect elements.

Also, those familiar with ANSYS/DesignSpace will note that the parabolic "bolt load" is available in that product. This provides the ANSYS user with a similar load application capability.

7. Summary:

This memo cannot present every feature of the structural surface effect elements in detail, but the author hopes that the above examples may aid the reader in the utilization of surface effect elements, in the event that he/she is unfamiliar with their capabilities. Later memos will provide more "advanced" functionality of the structural surface effect elements as well as their thermal counterparts.

To recap of the capabilities of the structural surface effect elements:

- The SURF153/154 elements can be generated directly via ESURF commands or meshed with LMESH/AMESH, overlaid on "regular" structural solid, planar, or shell elements.
- The use of pressure load keys for SURF154 are as follows: "1" designates a standard pressure load, "2" is pressure in the element x-axis, "3" is pressure in the element y-axis, "4" for pressure with tapering, and "5" for a pressure load with an arbitrary direction. SURF153 (2D element) has similar loading options
- Please refer to the Elements Manual for the various keyoptions and capabilities of these elements

These elements provide the ANSYS user with more flexibility and power in load application (in this case, application of pressures). However, as with any feature in ANSYS, the user new to surface effect elements is strongly encouraged to test modeling methods on simple models/examples before utilizing them for production work.

Sheldon Imaoka
Collaborative Solutions, Inc. (LA Office)
Engineering Consultant

ANSYS Tips and Tricks

"ANSYS Tips & Tricks" is provided for customers of Collaborative Solutions, Inc. (CSI) with active TECS agreements, distributed weekly in Adobe Acrobat PDF format via email. Unless otherwise stated, information contained herein should be applicable to ANSYS 5.4 and above, although usage of the latest version (5.6 as of this writing) is assumed. Users who wish to subscribe/unsubscribe can send an email to operator@csi-ansys.com with their name and company information with the subject heading "ANSYS Tips and Tricks subscription". Older issues may be posted on CSI's website at http://www.csi-ansys.com.

Corrections, comments, and suggestions are welcome and can be sent to operator@csi-ansys.com [they will be distributed to the appropriate person(s)]. While CSI engineers base their "Tips & Tricks" on technical support calls and user questions, ideas on future topics are appreciated. Users who wish to submit their own "ANSYS Tips and Tricks" are encouraged to do so by emailing the above address for more information.

XANSYS Mailing List

The xansys mailing list is a forum for questions and discussions of the use of ANSYS. As of 12/99, there are more than 900 subscribers with topics ranging from Structural, Thermal, Flotran, to Emag analyses, to name a few. Users are encouraged to subscribe to evaluate the usefulness of the mailing list for themselves. Also, either (a) using the mail program to filter [xansys] messages or (b) using the "digest" option to receive one combined email a day is strongly recommended to minimize sorting through the volume of postings.

```
This list is for *ALL* users of the ANSYS finite element analysis program from
around the world. The list allows rapid communication among users concerning
program bugs/ideas/modeling techniques. This list is NOT affiliated with ANSYS,
Inc. even though several members of the ANSYS, Inc. staff are subscribers and
regular contributors.

To SUBSCRIBE: send blank email to xansys-subscribe@onelist.com
To unsubscribe send blank email to xansys-unsubscribe@onelist.com
Archived on http://www.infotech.tu-chemnitz.de/~messtech/ansys/ansys.html
ANOTHER archive on http://www.eScribe.com/software/xansys/
(A poor archive is also at http://www.onelist.com/archives.cgi/xansys)
```

CSI ANSYS Technical Support

Collaborative Solutions, Inc. is committed to providing the best customer support in our industry. Three people will be devoted to technical support from 8:00am to 5:00pm PST every working day. CSI customers with active TECS (maintenance) agreements may contact CSI by any of the following ways:

```
Phone:  760-431-4815 (ask for ANSYS technical support)
Fax:  760-431-4824
Web:  http://www.csi-ansys.com
E-mail:  firstname.lastname@csi-ansys.com
Anonymous ftp site: ftp://ftp.csi-ansys.com

CSI Engineers:
Karen Dhuyvetter
Greg Miller
Sean Harvey
Bill Bulat
Sheldon Imaoka  (LA)
David Haberman
Mike Rife       (CO)
Alfred Saad
Mike Twining
```