Date	June 24, 2000	Memo Number	STI52:000624
Subject	**ANSYS Tips & Tricks: Converting Pressures to Forces**		
Keywords	Structural: Pressures: Nodal Forces		

1. Introduction:

While pressure loads are recommended for most structural applications, there are various instances where equivalent nodal forces are required. These may include reduced or reduced modal superposition methods for harmonic or transient analyses.

This memo attempts to cover one method of converting pressure loads into equivalent, consistent nodal forces using an APDL macro.

2. Background Discussion:

Internally, pressures are converted into equivalent nodal forces in ANSYS. It is always recommended to use pressures for two main reasons:

1. Single concentrated nodal forces result in stress singularities in 2D and 3D SOLID and SHELL elements. Single nodal forces are acceptable for COMBIN (spring) and BEAM elements since the cross-sections are not modeled explicitly. However, in all other cases, as noted above, a stress singularity will result. This may introduce difficulties associated with results interpretation since the stresses around the concentrated load will be artificially high. Also, stress singularities may cause convergence problems in nonlinear analyses.
2. To avoid stress singularities, distributed loads are recommended. However, the distribution of loads on element faces are not necessarily intuitive. For example, if one refers to Ch 2.4.2 "Quadratic Elements (Midside Nodes)" of the ANSYS Modeling and Meshing Guide, Figure 2-3 (below) illustrates consistent loading for various elements:



As noted in the figure above, while lower-order elements split a pressure loading into simple fractions of the areas associated with a node, midside node elements require somewhat unintuitive load distributions, especially in the case of higher-order hexahedral and tetrahedral elements (subfigures b and c, respectively). This is due to the fact that the shape function for the edge is parabolic, and the force loading needs to be consistent with the underlying stiffness representation of the element. Moreover, even with lower-order elements, if the mesh density varies in a specific region, the appropriate force loading will be cumbersome to calculate by hand.

From the above discussion, it is evident that concentrated nodal loads are not desirable, yet the conversion of element pressures to nodal forces is not straightforward. As a result, if nodal forces are needed, as in the examples of reduced or reduced mode superposition harmonic and/or transient analyses, another method must be sought.

3. "SI_P2F" Macro:

The "SI_P2F" macro is listed below:

```
/nopr
/pmacro

!----------------------------------------
! Store current set of nodes and elements
! in component for future retrieval
!----------------------------------------
cm,TEMPELEM,elem
cm,TEMPNODE,node

!----------------------------------------
! Perform static analysis
!----------------------------------------
finish
/prep7
immed,0
finish

/solu
antype,static
nlgeom,off

!----------------------------------------
! Select areas with pressures
! Comment this out if you want to select
! pressure prior to execution of macro
!----------------------------------------
*msg,WARN
Please select the areas with pressure
asel,s,p
!----------------------------------------
! Selects nodes on area and puts in
! component PRESNODE
!----------------------------------------
nsla,s,1
cm,PRESNODE,node
*get,MYNODE,node,,count
!----------------------------------------
! Select only elements attached to nodes
! We will solve this subset of elements
!----------------------------------------
esln
nsle

!----------------------------------------
! Constrain all nodes and solve
!----------------------------------------
d,all,all
solve
finish

!----------------------------------------
! Get results from this analysis
!----------------------------------------
/post1
set,last
!----------------------------------------
! Retreive reaction forces and place in
! array NARRAY where second index
!   1 = node number
!   2 = FX
!   3 = FY
!   4 = FZ
! also multiply forces by -1 since
reaction = -force
!----------------------------------------
cmsel,s,PRESNODE
*get,NCNMAX,node,,num,max
*del,NREACT(1),,nopr
*del,NMASK(1) ,,nopr
*dim,NREACT,array,NCNMAX,3
*dim,NMASK ,array,NCNMAX
*vget, NMASK(1),node,1,nsel
*vmask,NMASK(1)
*vget, NREACT(1,1),node,1,rf,fx
*voper,NREACT(1,1),NREACT(1,1),mult,-1
*vmask,NMASK(1)
*vget, NREACT(1,2),node,1,rf,fy
*voper,NREACT(1,2),NREACT(1,2),mult,-1
*vmask,NMASK(1)
*vget, NREACT(1,3),node,1,rf,fz
*voper,NREACT(1,3),NREACT(1,3),mult,-1
finish

!----------------------------------------
! Get rid of displacements we applied
!----------------------------------------
/solu
cmsel,s,PRESNODE
esln
nsle
ddele,all,all
!----------------------------------------
! Get rid of pressure on area
!----------------------------------------
sfadele,all,,all
!----------------------------------------
! Apply nodal forces on area
!----------------------------------------
*vmask,NMASK(1)
f,(1:NCNMAX:1),fx,NREACT(1:NCNMAX:1,1)
*vmask,NMASK(1)
f,(1:NCNMAX:1),fy,NREACT(1:NCNMAX:1,2)
*vmask,NMASK(1)
f,(1:NCNMAX:1),fz,NREACT(1:NCNMAX:1,3)
!----------------------------------------
! Restore original selected set of nodes
!----------------------------------------
cmsel,s,TEMPNODE
cmsel,s,TEMPELEM
/pbc,f,1
nplot
finish

!----------------------------------------
! Clean up after ourselves
!----------------------------------------
cmdele,PRESNODE
cmdele,TEMPNODE
cmdele,TEMPELEM
*del,MYNODE
*del,NCNMAX
*del,NREACT
*del,NMASK
/gopr
```

The author has written a very simple macro (listed above) to covert pressure loads on areas (SFA) into equivalent nodal forces. The main idea behind this approach is as follows:

1. Select nodes and elements associated with pressure-loaded areas
2. Constrain all DOF for those selected nodes and elements, and solve subset.
3. Use *GET (or *VGET)[1] to obtain reaction forces at nodes where pressure is applied.
4. Delete pressure-loaded areas.
5. Apply negative of reaction force to nodes which had pressure loading on element face.

To make the macro run more efficiently, the author uses vector commands.[2] However, some vector commands used are not documented, so the user can easily modify the macro to take advantage of documented methods of performing the above actions.

After using this macro,[3] the selected pressure loads on areas are automatically converted into equivalent, consistent nodal forces. The user now does not have to worry about differing mesh density or lower-/higher-order elements. Because the reaction forces are being utilized, the nodal forces have the correct distribution for any given force.

For eigenvalue buckling analyses or for general nonlinear analyses, please remember that nodal forces always follow their original orientation whereas element pressures are “follower-forces,” following the normals of the element faces in large deflection analyses. If conversion of pressures to nodal forces is performed, the user must keep this in mind.

This methodology can be extended to convert more complex surface loads (such as those used with SURF153/154 surface effect elements), including parabolic distribution, shear loads, or pressures on projected areas only. Support for FE loads (SF, SFE) can be easily accounted for. Lastly, extension to heat transfer (HFLUX) and other disciplines are possible. Inclusion of these features are left as an exercise to the reader. Users with active TECS maintenance agreements can also contact CSI’s technical support line for assistance.

As with any new/untested method/macro, the user is encouraged to verify this macro on simpler models before using it on production models. The author has done some limited testing to ensure that this macro converts pressures to equivalent nodal forces correctly, but the user should always verify results himself/herself.

4. Conclusion:

A simple macro was provided to convert pressures to equivalent nodal loads. It is important to understand that conversion of pressures to nodal loads is not a trivial matter for any discipline (structural, heat transfer, etc.). The underlying mesh density and element shape function must properly be accounted for (i.e., “consistent” loading).

The attached macro performs this by fully constraining the model, then using the reaction forces to extract the equivalent nodal force representation of the pressures. Then, the macro applies the negative of the reaction force on master DOF for reduced or reduced mode superposition analyses. This may also be used for regular mode superposition or substructure analyses in lieu of scaling load vectors.

Sheldon Imaoka
Collaborative Solutions, Inc. (LA Office)
Engineering Consultant

[1] Starting from 5.5, *GET of reaction forces has been documented. However, *VGET of reaction forces is still undocumented at 5.6. The user may modify the macro to use *GET in a *DO loop instead of *VGET, if the use of documented commands is desired.

[2] Please refer to CSI’s Tip of the Week on “Vector Commands and Functions” for more information.

[3] If the user is unfamiliar with macros, please refer to CSI’s Tip of the Week on “APDL Macros” by M. Rife on details of how to use macros in ANSYS.

ANSYS Tip of the Week

"ANSYS Tip of the Week" (TOTW) is provided for customers of Collaborative Solutions, Inc. (CSI) with active TECS agreements, distributed weekly in Adobe Acrobat PDF format via email. Unless otherwise stated, information contained herein should be applicable to ANSYS 5.4 and above, although usage of the latest version (5.6 as of this writing) is assumed. Users who wish to subscribe/unsubscribe or to view older TOTW archives can visit

`http://www.csi-ansys.com/tip_of_the_week.htm`

Corrections, comments, and suggestions are welcome and can be sent to operator@csi-ansys.com [they will be distributed to the appropriate person(s)]. While CSI engineers base their TOTW on technical support calls and user questions, ideas on future topics are appreciated. Users who wish to submit their own TOTW are encouraged to do so by emailing the above address for more information.

XANSYS Mailing List

The XANSYS mailing list is a forum for questions and discussions of the use of ANSYS. As of 04/00, there are more than 1000 subscribers with topics ranging from Structural, Thermal, Flotran, to Emag analyses, to name a few. Users are encouraged to subscribe to evaluate the usefulness of the mailing list for themselves. Also, either (a) using the mail program to filter [xansys] messages or (b) using the "digest" option to receive one combined email a day is strongly recommended to minimize sorting through the volume of postings.

```
This list is for *ALL* users of the ANSYS finite element analysis program from around the
world. The list allows rapid communication among users concerning program
bugs/ideas/modeling techniques. This list is NOT affiliated with ANSYS, Inc. even though
several members of the ANSYS, Inc. staff are subscribers and regular contributors.

To SUBSCRIBE: send blank email to xansys-subscribe@onelist.com
To unsubscribe send blank email to xansys-unsubscribe@onelist.com
Archived on http://www.infotech.tu-chemnitz.de/~messtech/ansys/ansys.html
ANOTHER archive on http://www.eScribe.com/software/xansys/
(A poor archive is also at http://www.onelist.com/archives.cgi/xansys)
```

CSI ANSYS Technical Support, Training, & Mentoring

Collaborative Solutions, Inc. is committed to providing the best customer support in our industry. Three people will be devoted to technical support from 8:00 a.m. to 5:00 p.m. PST every working day. CSI customers with active TECS (maintenance) agreements may contact CSI by any of the following ways:

```
Phone: 760-431-4815                                 WWW: http://www.csi-ansys.com
Fax:   760-431-4824                                 FTP:  ftp://ftp.csi-ansys.com
E-mail: firstname.lastname@csi-ansys.com
```

CSI Engineers:

Karen Dhuyvetter	Greg Miller	Sean Harvey	Alfred Saad
Bill Bulat	Sheldon Imaoka	David Haberman	Mike Rife

CSI believes strongly in the value of training and mentoring to help make customers successful using ANSYS. Training classes are usually 2-3 days in duration and provide instruction on various topics, including structural nonlinearities, heat transfer, and dynamics. Mentoring sessions involve working with a CSI engineer one-on-one on specific projects. These sessions help reinforce applicable subject matter covered in training classes or help ensure that the customer is using ANSYS most efficiently and effectively.

Training class schedules are posted at: `http://www.csi-ansys.com/training.htm`

Please contact your account manager for more details on training, mentoring, consulting, and other CSI services.