# Fast Nonlinear Transient Thermal Solver

**Beta Feature – ANSYS Release 5.6**

**User Guide**

## Overview

The ability to quickly solve large nonlinear transient thermal problems has been a priority development item of our user community. We have implemented a fast solver algorithm to address this issue. The solver option is a *BETA* release feature at 5.6 that will become a fully released feature at 5.7. Customers may experiment with the new solver algorithm at the first commercial release of 5.6. (Note: Beta 5.6 releases do not contain the full functionality implemented in the commercial release version).

## Solver Overview

The new solver strategy (called the Selective Reform Strategy) incorporates many changes to the present (Full Reform Strategy) algorithm. Significant speed-up has been realized without compromise to solution accuracy. Highlights of the new solver strategy are;

1. Solver matrices [K,C] are now assembled and stored in memory. The matrices are now independent of the time increment.
2. Temperature dependent material properties will cause a matrix reform only if a significant change in the property is detected (user-controlled feature).
3. The time-stepping algorithm now also considers changes in material properties. Equilibrium iterations have been removed.
4. The ICCG solver has been enhanced to make use of the previous solution as a starting point for the next solution, reducing the number of iterations. An undocumented JCGX solver has similarly been enhanced.
5. A Fast Table has been developed for material property evaluation.
6. Tabular boundary condition efficiency has been enhanced.

## Solver Commands

A new command has been implemented to activate the new solver strategy

THOPT, Option, ReformTol, nTabPoints, TempMin, TempMax

Option

NONLIN – Use Full Reform solver strategy (default)
QUASI – Use Selective Reform strategy (new fast solver)
LINEAR – Do not reform element matrices (new)

ReformTol – Material property change tolerance (expressed as a fraction) for matrix reformation. Matrix reformation will occur if any material property change between successive iterations is greater than the tolerance. Default value is .05. (For example, by default, if a material property changes by 5% between

iterations, the solution matrices will be reformed). Used only for the QUASI solver option.

NTabPoints – The number of temperature points for the Fast Material Table. Defaults to 64. All input temperature dependent material property tables will be converted into a single table with equal temperature increments between TempMin and TempMax. Used only for the QUASI option.

TempMin, TempMax – Minimum and Maximum temperatures for the Fast Material Table. Temperature values should span the values input by the individual material property tables (MPTEMP, MPDATA commands).

The following command must also be input to set up memory storage

DIRECT, ON

The following solver options are supported for the new Selective Reform solver strategy (QUASI);

EQSLV, ICCG
EQSLV, JCGX (undocumented)

Solver Requirements and Limitations

The Selective Reform strategy (QUASI) will require more memory than the Full Reform strategy (NONLIN) in order to store the matrices. As a general rule, 1 GigaByte of memory is required for every 1 Million degrees of freedom.

Surface to surface radiation is presently not supported using the Selective Reform strategy (QUASI). Use the NONLIN strategy if radiation is required.

Solver Accuracy

Initial studies indicate that solution accuracy is not compromised by using the Selective Reform Strategy. For auto time-stepping, the Selective Reform Strategy usually takes more substeps than the existing Full Reform Strategy. However, the Selective Reform Strategy does not perform equilibrium iterations at each time point. This fact in combination with a more efficient matrix formation and solvers is showing significant speed-up in the overall problem solution.

Sample Problem Input File

```
/BATCH,LIST
/PREP7
/show,file
/title, Nonlinear Transient Thermal: Beta Fast Solver Example
/com
/com,   A thick-walled vessel is initially at 100 deg.
/com,   A fluid is flowing through the vessel.  At time 0+, the outside
/com,   surface of the vessel is subject to convection with a bulk temp
/com,   of 600.  The internal fluid convects with the vessel walls and
/com,   extracts heat from the vessel.
/com,
/com,   The model is parametric, define the number of solid elements
/com,   desired.
/com,
/com,   The problem tests the undocumented command THOPT
/com,
/com,   THOPT, [NONLIN | QUASI | LINEAR], ReformTol, nTabPoints,TMin,
/com,    TMax
/com,       NONLIN - The current Rev 5.6 solution
/com,       QUASI - Selective reformation based on ReformTol
/com,       LINEAR - No reformation
/com,
/com,       ReformTol - Property change for Matrix Reformulation (.05
/com,        default)
/com,       nTabPoints - Number of points in Fast Table (64 default)
/com,       TMin,TMax - Min and Max temperatures for Tables
/com,
/com.       Note:  Time stepping strategy will be based on combination
/com,        of property change and Rayleigh strategy.
/com,
/com,*************************************************************

n=600       ! Number of solid elements  (multiple of 200)

/com,*************************************************************

et,1,55,,,1                     ! 2-d thermal solid - axisymmetric
et,2,116,1,1                    ! Thermal-flow element
et,3,151,,,1,1,1,0              ! Surface-effect element
keyopt,3,8,2
r,1,(1/12),.00545415            ! Diameter, Flow area

/com, Fluid (properties are fictitious)
mptemp,1,0,50,100,150,200,300
mpdata,kxx,1,1,.08,.10,.12,.14,.17,.23   ! BTU/hr-ft-F
mp,dens,1,1.4377E-10                     ! lbf-hr**2/ft**4
mpdata,c,1,1,.5e8,.7e8,.9e8,1.2e8,1.4e8,1.6e8
mp,visc,1,1.17418E-10                    ! lbf-hr/ft**2

/com,  Pipe (properties are fictitious)
mptemp,1,100,200,300,400,500,600
mpdata,kxx,2,1,50,75,100,125,150,175
mpdata,c,2,1,.2,.25,.3,.35,.4,.45
mp,dens,2,4.053e-7
```

```
len=n/200
siz=n/100

k,1
k,2,,len
type,2
real,1
l,1,2                   ! Fluid line
esize,,siz
lmesh,all
cm,fluid,line                ! Create line component
ir=.05                       ! Pipe inner radius (ft)
or=.50                       ! Pipe outer radius (ft)
rectng,ir,or+ir,0,len        ! create pipe wall
type,1
mat,1
esize,.05
amesh,1                      ! pipe wall

lsel,s,loc,x,ir
cm,surf,line                 ! inner pipe wall line component
allsel,all
type,3
real,2
/psymb,xnode,1
lfsurf,'surf','fluid'        ! mesh surface elements, connect to fluid
elements
allsel,all
esel,s,type,,3
sfe,all,,conv,,2.7             ! Equivalent heat transfer coefficient
esel,s,type,,2
sfe,all,,hflux,,1.132e-8     ! Flow rate in pipe  lbf/hr-ft
esel,all
dk,1,temp,100
lsel,s,loc,x,or+ir
sfl,all,conv,.8,,600         ! Convection to ambient
tunif,100
finish
/solu
antyp,trans
thopt,nonlin                 ! Use full reform strategy
time,.10
deltim,.00005,.00001,.05
autots,on
kbc,1
outres,all,none
outres,nsol,all
eqslv,iccg
*get,t1,active,,time,wall
solve
*get,t2,active,,time,wall
fini
s1=node(ir,0,0)
s2=node(or+ir,0,0)
/post26
nsol,2,2,temp                ! Exit fluid node
```

```
nsol,3,s1,temp                ! Entrance wall inner node
nsol,4,s2,temp                ! Entrance wall outer node
/com,
/com, ******** Full Reform Solution ******************************
prvar,2,3,4
/title, Full Reform Solution
plvar,2,3,4
/com,************************************************************
finish

/solu
antyp,trans
time,.10
deltim,.00005,.00001,.05
autots,on
kbc,1

thopt,quasi,.05,61,0,600 ! 5% reform strategy, 61 table points, 0<T<600
direct,on                ! Set up memory storage

outres,all,none
outres,nsol,all
eqslv,iccg
*get,t3,active,,time,wall
solve
*get,t4,active,,time,wall
/post26
nsol,2,2,temp
nsol,3,s1,temp
nsol,4,s2,temp
/com, *****Selective Reform Solution ******************************
prvar,2,3,4
/title, Selective Reform Solution
plvar,2,3,4
/com,
delt=t2-t1             ! compute elapsed solver times
delt2=t4-t3
/com,************************************************************
/com,
/com,******* Full Reform Elapsed Solution Time ********************
*status,delt
/com,**** Selective Reform Elapsed Solution Time ******************
*status,delt2
finish
```