Date April 9, 2000 Memo Number STI38:000409A
Subject **ANSYS Tips & Tricks: Thermal Surface Effect Elements**
Keywords Surface Effect: Thermal

1. Introduction:

Surface effect elements are extremely useful tools in both structural and thermal analyses. These elements provide the user with more flexibility in applying loads to their model. This memo is part 2 in a series which hopes to provide users with more information on the structural and thermal surface effect elements, SURF151-154 in ANSYS 5.5 and 5.6 (similar to SURF19 and SURF22 in ANSYS 5.4).

In this memo, the basics of thermal surface effect elements will be reviewed.

2. Background Discussion:

Surface effect elements have no physical properties. These elements are generally used for loading purposes only. They are overlaid like a "skin" on thermal (and structural) element faces.

In most thermal applications, only one surface load (SFx) can be applied on any given element face. Also, "simple" radiation on faces cannot be applied with the SFx family of commands.

On the other hand, thermal surface effect elements SURF151/152 allow the user to define multiple heat loads on surfaces, including radiation loads. A summary of the features of SURF151/152 are listed below:

- Application of convection loads, with or without an extra node to represent bulk temperature.
- Application of heat flux or heat generation.
- Application of radiation on load with view factor input as real constant or with cosine effect, using an extra node.
- Coupling with FLUID116 to account for bulk temperature rise.
- Use of subroutines USRSURF116 and USERCV for additional functionality.

With the above capabilities, multiple surface effect elements can be used to apply various types of heat loads on any element face, and postprocessing of these individual heat sources can be performed (via ETABLE).

3. Example of Generating Surface Effect Elements:

The SURF151/152 elements can be used with any lower- and higher-order thermal element, with the exception of axisymmetric-harmonic elements (PLANE75, PLANE78). These surface effect elements are generated on top of existing element faces via ESURF, LMESH/AMESH, or AFSURF/LFSURF commands (or direct generation).

For example, the below input file illustrates the two methods of generating surface effect elements:

```
! ESURF elements
/prep7
et,1,90
et,2,152
block,,10,,10,,10
vatt,1,1,1
esize,1
vmesh,all
asel,s,loc,y,10
nsla,s,1     ! select nodes for ESURF
esel,all     ! select elements for ESURF
type,2
esurf
```

```
! AMESH elements
/prep7
et,1,90
et,2,152
block,,10,,10,,10
vatt,1,1,1
esize,1
vmesh,all
asel,s,loc,y,10
type,2
amesh,all
```

As noted in the example, the left-hand side generates surface effect elements via the ESURF command, which directly generates them from selected nodes. ANSYS looks for selected nodes and elements on which to overlay surface effect elements. The selected nodes tell ANSYS which nodes to use to generate surface effect elements. The selected elements tell ANSYS which element faces to use (corresponding to the selected nodes) and how to orient the element z-axis (pointing away from solid element). Usually, selecting the nodes is most important, and the user can issue "ESEL,ALL" for the elements – ANSYS will not generate SURF151/152 on all the elements but only those faces whose nodes are selected.

The right-hand input shows another way of generating surface effect elements. Because the volume is meshed first, ANSYS knows how to orient the surface effect elements (element z-axis pointing away from volume). Unlike the ESURF command, AMESH allows the surface effect elements to retain associativity with the areas, so instead of EDELE, the user must issue ACLEAR to delete the surface effect elements.

Neither of the above methods is necessarily "better" than the other; it is simply a matter of preference for the user in generating surface effect elements.

For coupling SURF151/152 with FLUID116, two additional generation commands are provided, AFSURF and LFSURF. These commands generate surface elements on the surface of existing solid elements and assign the extra node as the closest fluid element node.

```
/graph,power
/view,1,1,2,3
/psym,xnode,1 ! show extra node (purple)
/pnum,type,1
/num,1
/auto
/prep7
et,1,70       ! lower-order brick
et,2,152      ! surface effect element
keyopt,2,4,1  ! use lower-order option
keyopt,2,5,1  ! use extra node
et,3,116      ! thermal-fluid pipe

k,1,0,15,5
k,2,10,15,5
l,1,2
type,3
esize,1
lmesh,all
cm,L_FLUID,line

block,,10,,10,,10
type,1
esize,1
```



```
vmesh,all
asel,s,loc,y,10
cm,A_SHELL,area

allsel,all
type,2
afsurf,'A_SHELL','L_FLUID'
allsel,all
eplot
```

The xFSURF family of commands provide a very useful, quick way to generate surface effect elements between the FLUID116 elements and thermal solids. For more information, please refer to the Commands Manual for LFSURF & AFSURF. Include in the online help description is the location of the equivalent xFSURF GUI menu item.

4. Example of Convection and Heat Flux on Face:

Below is a simple example of a bar with one end constrained (TEMP=20) and another end with both heat flux and convection loading, using surface effect elements.

```
finish
/clear

WIDTH    = 0.01
ELEMSIZE = WIDTH/5

/graph,power
/triad,lbot
/view,1,1,2,3
/auto
/pnum,type,1
/num,1
/psym,xnode,1
/prep7
et,1,70
et,2,152
keyopt,2,4,1 ! lower-order
keyopt,2,5,0 ! no extra node
keyopt,2,8,1 ! use heat flux only
et,3,152
keyopt,3,4,1 ! lower-order
keyopt,3,5,1 ! use extra node, bulk temp
keyopt,3,8,2 ! use convection only

mp,kxx ,1,156

block,,WIDTH,,10*WIDTH,,WIDTH
esize,ELEMSIZE
vmesh,all

asel,s,loc,y,10*WIDTH
nsla,s,1
esel,s,type,,1
type,2
esurf
n,1e6,0.5*WIDTH,15*WIDTH,0.5*WIDTH
esel,s,type,,1
type,3
esurf,1e6
finish

/solu
asel,s,loc,y,0
nsla,s,1
d,all,temp,20

esel,s,type,,2
sfe,all,1,hflux,,10/(WIDTH*WIDTH)

esel,s,type,,3
sfe,all,1,conv,1,20

nsel,all
d,1e6,temp,200

allsel,all
solve
finish

/post1
set,last
esel,s,type,,1
nsle
plnsol,temp

esel,s,type,,2 ! get surf eff only
esel,a,type,,3
etable,HGTOT ,smisc,1 ! should be 0
etable,HFCTOT,smisc,2 ! hflux and conv
etable,HRTOT ,smisc,3 ! should be 0
pretab
```

The resulting temperature profile is shown on the right. Using element tables (ETABLE), one can verify the heat flux and convection loads on the surface effect elements.



For the applied heat flux, 10 W was applied on the area. As a result, each element (25 elements total) should have 0.4 W, and this is reflected in the results (PRETAB).

For the convection load, we find that the total heat flow due to convection is $hA\Delta T$ or 0.22886 W. For 25 elements, the heat flow rate is 9.1544e-3 which is also verified in the results (PRETAB). [Note that the heat flow is negative since it is leaving the surface effect element]

These elements not only allow for multiple heat loads on the same face (using two sets of surface effect elements, in this specific case), but one can use ETABLE to selectively postprocess the amount of heat leaving/entering each element. This helps confirm values with hand calculations as well as determining which heat path is most critical in a system.

5. Example of Heat Generation:

The following is an example of using SURF152 for surface heat generation. A thickness is given to SURF152, and a heat generation rate (Power/Volume) is applied on SURF152 only.

```
THICKNSS = 0.001
WIDTH    = 0.01
HEIGHT   = 0.1
S_VOLUME = THICKNSS*WIDTH**2

/prep7
et,1,70
et,2,152
keyopt,2,4,1
keyopt,2,5,0
keyopt,2,8,0

mp,kxx ,1,156

r,2,
rmodif,2,7,THICKNSS

block,,WIDTH,,HEIGHT,,WIDTH

esize,WIDTH/5

vmesh,all

asel,s,loc,y,HEIGHT
nsla,s,1
type,2
real,2
esurf
finish

/solu
asel,s,loc,y,0
nsla,s,1
d,all,temp,20

esel,s,type,,2
nsle,s,1
bfe,all,hgen,,10/S_VOLUME
allsel,all
solve
finish

/post1
set,last
esel,s,type,,1
nsle
plnsol,temp

esel,s,type,,2
etable,HGTOT ,smisc,1
etable,HFCTOT,smisc,2
etable,HRTOT ,smisc,3
pretab
prrsol
```

By using the element tables (PRETAB), we see that only heat generation is used on the element. The heat generation is 10W divided by the 25 elements, or 0.4W, which is verified in the element table.

If we consider Fourier's Law, the answer to this simple problem is:

$$q = kA\frac{\Delta T}{\Delta x}$$

$$T_e = q\Delta x / kA + T_b$$

$$= 64.1 + 20$$

The resulting answer is that the end should be 84.10256 °C, as shown in the temperature profile at right. This verifies that the heat generation is applied as expected.



To verify that the thickness of the SURF152 is not physical (i.e., does contribute to the thermal field calculations), one can change the THICKNSS variable from 0.001 to 10 and rerun the analysis. The same results should be shown since, as noted in Section 2, surface effect elements have no physical significance. The thickness of the surface effect elements are used in mass and heat generation calculations only; the surface effect elements do not "conduct" heat as a regular thermal shell element does.

This feature of the surface effect elements make them very attractive. In some instances, it may be easier to apply heat generation instead of heat flux, but both options are available with SURF151/152.

6. Radiation with Cosine Effect:

Surface effect elements provide simple radiation boundary conditions, i.e., node-to-surface radiation. The form factor can be user-defined (as in a real constant) or calculated from the cosine effect, using the position of the extra node.[1]

```
WIDTH    = 0.01

/graph,power
/triad,lbot
/view,1,1,2,3
/auto
/pnum,type,1
/num,1

/prep7
et,1,87
et,2,152
keyopt,2,4,0
keyopt,2,5,1
keyopt,2,9,3 ! radiation, cosine eff

mp,kxx ,1,156
mp,emis,2,0.8

r,2,
rmodif,2,2,5.67e-8
toffst,273.15

block,,WIDTH,,WIDTH,,WIDTH

esize,WIDTH/7
vmesh,all

asel,s,ext
nsla,s,1
n,1e6,2*WIDTH,2*WIDTH,WIDTH/2
type,2
real,2
mat,2
esurf,1e6
finish

/solu
eqslv,jcg
nsubst,2,10,2

asel,s,loc,x,0
!asel,a,loc,y,0
!asel,a,loc,z,0
nsla,s,1
d,all,temp,20

nsel,s,node,,1e6
d,all,temp,1000

allsel,all
solve
finish

/post1
set,last
esel,s,type,,1
nsle
plnsol,temp

esel,s,type,,2
nsle
etable,HGTOT ,smisc,1
etable,HFCTOT,smisc,2
etable,HRTOT ,smisc,3
pretab
etable,FORMF ,nmisc,16
pletab,FORMF
```

The above example demonstrates some of the radiation capabilities of the thermal surface effect elements. A simple block with 20 °C imposed at one end is subject to radiation from a “space” node which is at 1000 °C. Radiation form factors are calculated from the cosine of the normal to the extra (space) node. Any “negative” view factors are neglected and set to zero (KEYOPT(9)=3).

The average form factors for each element are stored in an element tables and plotted at right. The extra space node lies at a corner edge of the block, so four of the sides of the cube should have no radiation (shown in blue). Simple hand calculations verify that the form factors should lie between 0.5 and 0.7.



[1] See Theory Manual, Ch. 14.152.3 for information on cosine effect calculation.

7. Conclusion/Recommendations:

Thermal surface effect elements provide greater flexibility than standard SFx commands in heat transfer problems, both for multiple surface load application and postprocessing.

As noted from the input files, it is easiest to select surface effect elements, then apply finite element loads to the SURF151/152 elements (on face 1). While one can use solid model loads, one must use care to ensure that solid model loads are not transferred to the underlying SOLID (or SHELL) elements, so the author recommends applying loads directly on the elements to ensure that no problems arise.

To verify the applied loads, besides viewing them with /PSF and /PBF plot control commands, one can postprocess these results after solution. Using the SURF151/152 ETABLE values allow a user to (a) verify that correct heat flow rates are applied to each type of surface effect elements and (b) determine what are the most significant heat paths in one's system for design considerations.

As in the case with structural surface effect elements, there are many keyoptions for these thermal elements. The user is asked to create simple models to familiarize oneself with thermal surface effect elements prior to including them in production models.

Future memos are planned to discuss the usage of FLUID116 with SURF151/152 to model simple fluid-flow elements and convection of the fluid to thermal solids. This memo only serves as an introduction to some of the more commonly-used options for thermal surface effect elements.

Sheldon Imaoka
Collaborative Solutions, Inc. (LA Office)
Engineering Consultant

ANSYS Tip of the Week

"ANSYS Tip of the Week" (TOTW) is provided for customers of Collaborative Solutions, Inc. (CSI) with active TECS agreements, distributed weekly in Adobe Acrobat PDF format via email. Unless otherwise stated, information contained herein should be applicable to ANSYS 5.4 and above, although usage of the latest version (5.6 as of this writing) is assumed. Users who wish to subscribe/unsubscribe or to view older TOTW archives can visit

`http://www.csi-ansys.com/tip_of_the_week.htm`

Corrections, comments, and suggestions are welcome and can be sent to operator@csi-ansys.com [they will be distributed to the appropriate person(s)]. While CSI engineers base their TOTW on technical support calls and user questions, ideas on future topics are appreciated. Users who wish to submit their own TOTW are encouraged to do so by emailing the above address for more information.

XANSYS Mailing List

The XANSYS mailing list is a forum for questions and discussions of the use of ANSYS. As of 04/00, there are more than 1000 subscribers with topics ranging from Structural, Thermal, Flotran, to Emag analyses, to name a few. Users are encouraged to subscribe to evaluate the usefulness of the mailing list for themselves. Also, either (a) using the mail program to filter [xansys] messages or (b) using the "digest" option to receive one combined email a day is strongly recommended to minimize sorting through the volume of postings.

```
This list is for *ALL* users of the ANSYS finite element analysis program from around the
world. The list allows rapid communication among users concerning program
bugs/ideas/modeling techniques. This list is NOT affiliated with ANSYS, Inc. even though
several members of the ANSYS, Inc. staff are subscribers and regular contributors.

To SUBSCRIBE: send blank email to xansys-subscribe@onelist.com
To unsubscribe send blank email to xansys-unsubscribe@onelist.com
Archived on http://www.infotech.tu-chemnitz.de/~messtech/ansys/ansys.html
ANOTHER archive on http://www.eScribe.com/software/xansys/
(A poor archive is also at http://www.onelist.com/archives.cgi/xansys)
```

CSI ANSYS Technical Support, Training, & Mentoring

Collaborative Solutions, Inc. is committed to providing the best customer support in our industry. Three people will be devoted to technical support from 8:00 a.m. to 5:00 p.m. PST every working day. CSI customers with active TECS (maintenance) agreements may contact CSI by any of the following ways:

```
Phone:  760-431-4815                              WWW: http://www.csi-ansys.com
Fax:    760-431-4824                              FTP:  ftp://ftp.csi-ansys.com
E-mail: firstname.lastname@csi-ansys.com
```

CSI Engineers:

Karen Dhuyvetter	Greg Miller	Sean Harvey	Alfred Saad
Bill Bulat	Sheldon Imaoka	David Haberman	Mike Rife

CSI believes strongly in the value of training and mentoring to help make customers successful using ANSYS. Training classes are usually 2-3 days in duration and provide instruction on various topics, including structural nonlinearities, heat transfer, and dynamics. Mentoring sessions involve working with a CSI engineer one-on-one on specific projects. These sessions help reinforce applicable subject matter covered in training classes or help ensure that the customer is using ANSYS most efficiently and effectively.

Training class schedules are posted at: `http://www.csi-ansys.com/training.htm`

Please contact your account manager for more details on training, mentoring, consulting, and other CSI services.