THE UNIVERSITY OF ADELAIDE
Department of Mechanical Engineering

INTERNAL REPORT

Coupled Structural - Acoustic Analysis Using ANSYS

Carl Howard
8th March, 2000

Number of Pages:	15 pages
Document Number:	
Filename:	fsi-ansys.doc

CONTENTS

Introduction

This document describes procedures for conducting a coupled structural acoustic analysis using ANSYS. Numerical results are presented for an analysis of an infinite duct that is divided by a thin panel. Several types of elements were trialled in the analyses and it was found that the shell elements (SHELL63), had the least computational demand.

The purpose of conducting these trials on a simple model of a duct, was to develop the correct procedures that will be used later to analyse more complex models.

The Structure

The figure below shows the system under investigation.



An infinite duct with rigid walls was divided by a simply supported aluminum panel with the following properties:

Young's Modulus	70.3 GPa
Poisson's Ratio	0.35
Density	2700 kg / m^3
Thickness	3.376mm
Damping Loss Factor	$1x10^{-4}$ at 100Hz
Dimensions	0.5m x 0.5m

The ends of the duct were anechoically terminated. This is achieved by specifying that the elements at the end of the duct have infinite absorption in their material properties.

Background

The xansys newsgroup (`http://www.escribe.com/software/xansys/index.html`) describes that the use of shell elements with fluid structure interfaces on both sides will not work because each node in the shell element will have the same pressure on each side of the shell. An alternative is to define two layers of shells that are coincident and then couple the displacement degrees of freedom (DOF) and adjust the material properties appropriately to represent a single panel.

Other suggestions that I received were to use solid elements and not to be concerned with the poor aspect ratio warnings as ANSYS is a 64 bit program and will not suffer severe numerical degradation in the results. The results of my investigations presented here demonstrate that this suggestion is incorrect.

Static and Modal Analyses

Single Layer Shell Model

The aim of this section is to demonstrate the equivalency between a single layer of shell elements and a double layer of shell elements with modified material properties. The shell elements used in these analyses are type `shell63`.

A model was created with dimension 0.5m x 0.5m meshed with 25 shell elements as shown below.



The boundary conditions on the edges of the model were assigned simple support conditions. A point force was applied at x=0.2m y=0.3m of F_z=10N.

A static analysis and a modal analysis were performed on this model. The lumped mass approximation was turned on for the solution processing. The static analysis calculated the displacement of the panel beneath the point load and the modal analysis calculated the first 5 resonance frequencies.

	ANSYS	Exact	% Difference
Displacement (m)	0.104749e-3	?	
Resonance Frequencies (Hz)	65.849	66.71	-1.29
	162.00	166.77	2.86
	162.00	166.77	2.86
	250.08	266.84	6.28
		333.55	

The exact theoretical solution for the displacement was not calculated. These results will be called the baseline results in this report. Several other models will be examined and the results from the analyses will be compared to these baseline results.

Double Layer Shell Model

As described in the background section, one way to perform a coupled structural acoustic analysis using shell elements is to define two layers of shell elements and couple the appropriate displacement DOF's. To keep the same mechanical behaviour of the plate for the double layer of shells as the single layer of shells, the material properties must be altered. The Young's modulus and the density of the material must be divided by 2 and the thickness of the panel remains unaltered. To couple the nodes together between the two coincident panels, select the appropriate nodes in each panel and issue the commands:

```
! Couple coincident nodes
CPINTF,UX,0.0001,
CPINTF,UY,0.0001,
CPINTF,UZ,0.0001,
CPINTF,ROTX,0.0001,
CPINTF,ROTY,0.0001,
CPINTF,ROTZ,0.0001,
```

Check that the elements in the top layer have separate nodes to the elements in the bottom layer. Assuming that the top layer is type 10 and the bottom layer is type 11, then issue the following commands,

```
esel,s,type,,10   !select the top layer
nsle,s,1          !select the nodes for these elements
esln,s            !select any elements attached to these elements
esel,u,type,,10   !unselect the top layer elements
nsle,s,1          !select nodes attached to remaining elements
elis              !check that type 11 elements are not listed
```

There should not be any type 11 elements in this list. The material properties should be changed so that the Young's modulus and density are divided by 2.

```
!material properties
MP,EX,10,70.3e9/2 !Young's modulus
MP,DENS,10,2700/2 !density of aluminium
```

A static and modal analysis was conducted on this model and the results are listed below:

	Single Layer	Double Layer	% Difference
Displacement (m)	0.104749e-3	0.104749e-3	0
Resonance Frequencies (Hz)	65.849	65.849	0
	162.00	162.00	0
	162.00	162.00	0
	250.08	250.08	0

Identical results were obtained for the single layer and the double layer with the modified material properties.

Solid Model 4 Elements Thick

Another model was created using `solid45` elements with the same number of divisions along each face as the shell element model. Four elements through the thickness of the panel were used. Simple supports were assigned to the nodes along the neutral axis of the panel. A suggestion was made in the xansys newsgroup that the poor aspect ratio of the elements should not affect the results, however this is clearly not the case.

	Single Layer	Solid45 4 elements thick	% Difference
Displacement (m)	0.104749e-3	0.161009e-4	-84.6%
Resonance Frequencies (Hz)	65.849	157.90	139.8
	162.00	495.90	206.1
	162.00	495.90	206.1
	250.08	942.33	276.8

Solid Model 1 Element Thick

Another suggestion was to use a single layer of `solid45` elements. Simple support conditions were assigned to the nodes along the top face of the volume. The results are almost the same as using 4 elements through the thickness and are also incorrect.

	Single Layer	Solid45 1 elements thick	% Difference
Displacement (m)	0.104749e-3	0.160955e-4	-84.6%
Resonance Frequencies (Hz)	65.849	157.98	139.9
	162.00	495.66	205.9
	162.00	495.66	205.9
	250.08	944.41	277.6

Improved Element Divisions

For interest, the models described previously were re-meshed with 50 divisions along each edge and re-analysed. The results are listed below:

	Exact	Single Layer	Double Layer	% Difference from Single Layer
Displacement	?	1.01790E-04	1.01790E-04	0.00
Resonance Frequencies	66.71045	66.70	66.70	0.00
	166.7761	166.74	166.74	0.00
	166.7761	166.74	166.74	0.00
	266.8418	266.71	266.71	0.00
	333.5523	333.45	333.45	0.00

	Exact	4 Elements through thickness	% Difference from Single Layer	1 Element through thickness	% Difference from Single Layer
Displacement	?	1.01816E-04	0.03	1.01816E-04	0.03
Resonance Frequencies	66.71045			66.64	-0.09
	166.7761			166.66	-0.05
	166.7761			166.66	-0.05
	266.8418			266.43	-0.10
	333.5523			333.45	0.00

These results show that to correctly model the panel using solid elements, more than 5 elements per edge are required.

Transmission Loss Analyses

Double Layer Shell Elements

The infinite duct described in the structure section was modelled in ANSYS using `fluid30` elements for the acoustic space and `shell63` elements for the structure. The structure was modelled using a double layer of coincident elements as described previously.

The acoustic elements in the duct were assigned 4 different element types:

```
! Upstream 3-D ACOUSTIC FLUID elements
! without Fluid Structure interface
ET,1,FLUID30,,1

! Upstream 3-D ACOUSTIC FLUID elements
! with Fluid Structure interface
ET,2,FLUID30

! Downstream 3-D ACOUSTIC FLUID elements
! without Fluid Structure interface
ET,3,FLUID30,,1

! Downstream 3-D ACOUSTIC FLUID elements
! with Fluid Structure interface
ET,4,FLUID30
```

The material properties for the acoustic fluid were defined as:

```
!material properties for typical air
R,1,20e-6         !reference pressure (defaults to 20e-6)
densair = 1.21
MP,DENS,1,densair !air density
MP,SONC,1,344     !sound speed in air
```

The elements at the end of the duct were altered to have a different material property compared to the rest of the acoustic elements.

```
!Material for air with absorptive ends
R,3,20e-6         !reference pressure (defaults to 20e-6)
MP,DENS,3,densair !air density
MP,SONC,3,344     !sound speed in air
MP,MU,3,1         !sound absorption
```

In this model the intersection between the upstream and downstream sections do not share common nodes. Hence the upstream and downstream sections are structurally and acoustically isolated. The upstream acoustic fluids are type 1 for the duct and type 2 for the fluid structure interface. The fluid structure interface is 1 element wide. The upstream structure is type 10 and the downstream structure is type 11. The downstream acoustic fluids are type 3 for the duct and type 4 for the fluid structure interface.

The fluid structure interface is assigned to the fluid elements (not the structure). To turn on the FSI flag, select the nodes that in contact with the structure and select the elements that are in contact with the structure. For example:

```
esel,s,type,,10   !select the upstream structure
nelem             !select the nodes attached to elements
esel,s,type,,2    !select the fsi elements
sf,all,fsi        !turn on the fsi flag
```

The elements at the ends of the duct can be changed to anechoic end conditions by following this example:

```
!put in absorptive (anechoic) ends
nsel,s,loc,z,5
nsel,a,loc,z,-5
enode
nelem
mat,3
real,3
emodif,all,mat,3
emodif,all,real,3
nsel,s,loc,z,5
nsel,a,loc,z,-5
esln,s
sf,all,impd,1
```

The acoustic elements that are not in contact with the structure should have their displacement DOF's fixed

```
esel,s,,type,,1   !upstream acoustic fluid
esel,a,type,,3    !downstream acoustic fluid
nelem
d,all,ux
d,all,uy
d,all,uz
```

The upstream and downstream panels are connected using coupling equations

```
esel,s,type,,10,11
nelem
CPINTF,UX,0.0001,
CPINTF,UY,0.0001,
CPINTF,UZ,0.0001,
CPINTF,ROTX,0.0001,
CPINTF,ROTY,0.0001,
CPINTF,ROTZ,0.0001,
```

A harmonic pressure of 1Pa was applied 2m upstream from the panel. Note that a 1Pa pressure is equivalent to $20\log_{10}(1/20e\text{-}6/\sqrt{2})$=90.97dB.

```
! apply a pressure loading of 1Pa
nsel,s,loc,z,2
d,all,pres,1
```

A harmonic analysis was conducted at 100Hz using the frontal solver. The acoustic fluid elements were selected and then the sound pressure level was plotted.

```
esel,s,type,,1,4
nelem
plesol,nmisc,4
```

The sound pressure level in the duct is shown in the figure below.



The sound pressure level in the downstream section is 69.0dB and the noise reduction across the panel is 22dB.

Poor Models

For reference, the solid models considered previously were also analysed at 100, 150, 200, 250, 300 and 350 Hz. The noise reduction across the panel is shown in the figure below:



The results are for a double layer of `shell63` elements, a single layer of `solid45` elements, 4 layers of `solid45` elements that were simply supported across the neutral axis and 4 layers of `solid45` elements that were simply supported on the upper face. The surprising result was that at the higher frequencies the results from the solid model compare favorably with the results from the double layer of shell elements, despite that the solid element models do not behave as a simply supported panel.

Separated Upstream and Downstream Models

An alternative method of analysing the fully coupled structural acoustic model is to divide the problem into two simpler cases. First consider the upstream section to be anechoically terminated on one end and a rigid wall termination on the other, instead of the movable structure. A harmonic analysis can be performed and the pressure on the rigid end wall can be determined. Second, the downstream section can model the fluid structure interaction between the panel and the downstream acoustic fluid. The pressure determined from the upstream model can be applied to the exterior of the structure (still requires a double layer of shell elements) and the interior shell layer can couple to the downstream acoustic fluid.

The figure below shows the sound pressure level in the duct for the first case, when the upstream structure was replaced with a rigid wall.



Note that the pressure in the downstream section is zero because it is not coupled with the upstream elements. The real part of the complex pressure in the duct adjacent to the rigid wall is shown below.



The pressure at all the nodes on the end wall was -1.14349Pa. The imaginary part of the complex pressure was zero. The measured pressure can be applied to a second model that includes the fluid structure interaction between the panel and the downstream fluid. Another panel is defined to be co-incident and coupled with the panel in the downstream section. The new panel uses the altered material properties. Note that the pressure is applied to the shell elements in the new panel using the command

```
sf,all,pres,-1.14349
```

The sound pressure level in the downstream section at 100Hz is shown below



The sound pressure level in the duct is 69.3dB, which means that the noise reduction across the panel was 21.7dB. The noise reduction obtained using the fully coupled analysis was 22dB, hence these two values are sufficiently close to each other for our modelling purposes.

Appendix A: Generating Editable Postscript Figures

The default postscript output from ANSYS is a bitmap encapsulated in postscript and cannot be easily edited using graphics software such as Corel Draw and Adobe Illustrator. To generate editable postscript images, the display type must be changed. The display types are listed below and a tick indicates that the postscript output can be edited.

Editable Postscript	Type	Description
✔	0	Basic display (no hidden or section operations)
✔	1	Section display (plane view) use `/cplane`
✔	2	Centroid hidden display (based on item centroid sort)
✔	3	Face hidden display (based on face centroid sort)
✔	4	Precise hidden display (like 3 but with more precise checking)
✔	5	Capped hidden display (same as combined 1 and 3 with model in front of section plane removed)
	6	Z-buffered display (like 3 but using software Z buffering)
	7	Capped Z buffered display (same as combined 1 and 6 with model in front of section plane removed)
	8	Qslice Z buffered display (same as 1 but the edge lines of the remaining 3-D model are shown)
✔	9	Qslice precise hidden display (like 8 but using precise hidden)

1. In ANSYS set up your screen to display your required output figure.
```
eplo
```
2. Select the desired display type
```
/type,1,2               !window 1, centroid hidden display
```
3. Change the output to a file
```
/show,filename,grph     !output written to filename.grph
```
4. Replot the image
```
/rep
```
5. Print any other screen outputs to this file
```
plesol,nmisc,4
```
6. Return the output to the screen
```
/show,x11
```

Start the display program, then enter the following commands

```
file,filename,grph
/show,pscr
pscr,color,2
```

Then enter the following commands with the appropriate numbers depending on your desired page orientation

	Portrait	Landscape
`pscr,tranx,`	40	540
`pscr,trany,`	200	100
`pscr,rotate,`	0	90
`pscr,scale,`	0.18	0.2

To print out all the plots type the command

```
plot,all
```

The directory will contain the encapsulated postscript files pscr**.grph from the screen captures.